

VISUALIZACIÓN DE PATRONES DE RADIACIÓN SÍSMICA CON *MATHEMATICA*

J. Frez C.

Depto. de Sismología, División de Ciencias de la Tierra, CICESE
Km 107, Carret. Tijuana-Ensenada, Ensenada, B.C., 22860, México
E-mail: jofrez@cicese.mx

RESUMEN

Se presentan programas de computación que utilizan el paquete *Mathematica* para visualizar tridimensionalmente la distribución de la amplitud de arribos sísmicos modelados para fuentes puntuales asociadas a fallas. El ángulo de salida y el acimut del rayo sísmico calculados en el hipocentro del sismo son las variables independientes de las distribuciones. Se maneja un medio infinito para las ondas P y S, así como un modelo simple de semiespacio lateralmente homogéneo (sensor en la superficie) para los primeros arribos P, SV y SH, en componentes verticales y horizontales. Para el segundo modelo, se incluye además el cálculo de la razón de amplitudes SV/P en componente vertical. La capacidad de visualización aquí documentada ayuda a entender la teoría matemática y a interpretar datos.

INTRODUCCION

La representación matemática de una falla como fuente sísmica puntual (dislocación de cizalle) tiene numerosas convenciones en la definición de sus parámetros, en particular, de sus ángulos. Además, si queremos tomar en cuenta las amplitudes de los arribos más conspicuos en los sismogramas, debemos considerar que la posición de los sensores está en o muy cerca de la superficie terrestre; ésto agrega más convenciones, por ejemplo las que definen el signo, positivo o negativo, en un desplazamiento sísmico. Varios autores unifican y normalizan símbolos y definiciones; entre ellos están Jarosch y Aboodi (1972), Herrman (1975), Simila (1983), Jost y Herrman (1989) y Pujol y Herrman (1990). Por otro lado, los capítulos 4 y 5 del texto de Aki y Richards (1980) son referencia obligada en este campo. Un problema menor es que la mayor parte de las figuras que tienen estas referencias son de funciones univariantes a pesar del uso de coordenadas tridimensionales. Este artículo ofrece herramientas para hacer visualizaciones que ayuden a la comprensión intuitiva de las ecuaciones que modelan el efecto combinado de fuente y estructura sísmicas. La utilización de paquetes de programación puede ser criticada cuando es hecha en forma mecánica (v. gr., Frez, 1998); la crítica tiene poco asidero cuando se trata de visualizar problemas directos donde es fácil evitar problemas tales como criterios de suavización poco controlados que los paquetes comerciales aplican al conectar valores calculados en mallas discretas. *Mathematica* (Wolfram, 1991) es una de las herramientas más populares para hacer cálculos numéricos, manipulación simbólica y gráficas en ciencias naturales y matemáticas (v. gr., Crandall, 1991; Cordero *et al.*, 1995); es de fácil acceso y manejo, además de que produce gráficas de excelente calidad. Una buena herramienta de visualización estimula la intuición y la interacción con los desarrollos puramente matemáticos. Considero que la programación, gráficas y discusión que se ofrecen en este artículo pueden ayudar a los estudiantes de Ciencias de la Tierra.

FUENTES Y AMPLITUDES SÍSMICAS

Las ecuaciones que representan el patrón de radiación, en el campo lejano, producido por una dislocación puntual en un medio infinito pueden ser encontradas en numerosos textos y artículos; ver, por ejemplo, Aki y Richards (1981). Las ecuaciones para registros en la superficie libre de un semiespacio homogéneo pueden formarse a partir de las ecuaciones 4.84, 4.85, 4.86 y 4.87, además de los resultados del ejercicio 5.6, en Aki y Richards (1981). La amplitud del desplazamiento sísmico en la superficie contiene el efecto del patrón de radiación y el de superficie, donde se agrega la contribución del campo reflejado al del incidente. El efecto de transmisión puede modelarse en forma muy elemental para un medio elástico lateralmente homogéneo a través de la constancia del parámetro de rayo p . No incluimos efectos de atenuación física y suponemos que la atenuación geométrica es similar para los rayos P y S, cosa que queda justificada en parte al usar un valor constante de la razón de velocidades V_p/V_s ($\sqrt{3}$ para material poissoniano). Las ecuaciones originales se han normalizado haciendo igual a 1.0 el factor $1/(4\pi\rho c^3)$, donde ρ es la densidad, c es la velocidad sísmica que corresponde (V_p , V_s) y los valores corresponden a la estructura en el foco sísmico.

Con todo ésto, podemos comparar las amplitudes de los arribos calculados de forma bastante realista. Las limitaciones de este modelo son evidentes aunque no impiden que los cálculos ayuden en el trabajo rutinario y sean una herramienta fácil de utilizar como primera opción (por ejemplo, en Kisslinger, 1980), es decir, antes de cálculos más complicados como el de sismogramas sintéticos realistas. Además, muestra un panorama global del campo sísmico con bajo costo computacional.

En este trabajo, he utilizado la versión 2.2 de *Mathematica* para computadoras personales (PC). La mayor parte de los cálculos fueron hechos en una PC de apenas 60 Mhz y 24 Mb de

memoria. La pérdida de precisión numérica no aparece en estos cálculos, aunque debe preocuparnos una eventual división entre cero o entre algún número pequeño así como la aparición de números complejos y de efectos no deseados que resultan de la discretización. Nos referiremos más adelante a estos tópicos en un caso concreto.

De los diferentes comandos para realizar la proyección en dos dimensiones de figuras tridimensionales, he utilizado consistentemente “ParametricPlot3D”, que es confiable y rápido en ejecución. Los comentarios en los programas están entre paréntesis con asterisco (* ... *), según la sintaxis de *Mathematica*. Algunos comandos aparecen comentados para dejar al usuario la posibilidad de utilizarlos. Por ejemplo, se puede experimentar con cambios del punto de vista para la proyección que se utiliza en las figuras. Es preferible utilizar escalas en los ejes para comparar amplitudes de diferentes arribos. Las variables independientes de las figuras son el ángulo de salida del rayo desde la fuente (ángulo polar, i_h) y la dirección de salida del rayo con respecto a la traza de la falla (ángulo acimutal, $\phi - \phi_s$). Entre los paréntesis anteriores, agrego la coordenada esférica que corresponde, donde $\phi - \phi_s$ es la diferencia entre el acimut de la traza de la falla y el de la salida del rayo (Figura 1). En las gráficas, he utilizado coordenadas cartesianas y, en figuras que ofrecen ejemplos de patrones de radiación, el eje X corresponde a la dirección de la traza de la falla. La Figura 1 ilustra las variables relacionadas con la descripción matemática del movimiento de una falla y del patrón de radiación sísmica. El plano {X,Y} corresponde al plano ecuatorial de la esfera focal, es decir, de una pequeña esfera alrededor del foco sísmico. En un procedimiento que es usual para determinar mecanismos focales con la polaridad del primer arribo P, los valores de amplitudes en la superficie se calculan sobre la esfera focal y esta posición es proyectada sobre el plano ecuatorial. Un rayo que emerge hacia arriba desde el foco tiene un ángulo de salida que es función positiva de la distancia epicentral; a medida de que aumenta la distancia, el ángulo tiende a 90° para posteriormente convertirse en un rayo buzante que sale por el hemisferio inferior de la esfera focal. En los programas que calculan amplitudes en la superficie de un semiespacio, las coordenadas en el plano ecuatorial son $x = \sin(i_h) \cos(\phi - \phi_s)$ y $y = \sin(i_h) \sin(\phi - \phi_s)$. De este modo, el módulo de un vector posición en ese plano dividido entre la velocidad en el foco corresponde al parámetro del rayo. Opcionalmente, se puede utilizar una proyección equiareal; el correspondiente cálculo de las coordenadas para este caso está comentado en el código entregado en la Programa 2. Los Programas 2 al 4 cuentan con un primer segmento donde se definen las variables de entrada; luego, se aplican las ecuaciones utilizando variables auxiliares; por último, se llama al comando ParametricPlot3D -o ParametricPlot- dentro del cual, entre otras cosas, se define el rango de las variables independientes. *Mathematica* permite obtener un archivo PostScript de cada dibujo. Las figuras que se consiguen en pantalla son a colores.




Figura 1. Variables y sistemas de coordenadas que se utilizan para describir el movimiento en una falla y el patrón de radiación. El origen se toma en el epicentro (el punto en la superficie de la Tierra que se encuentra verticalmente sobre la fuente sísmica). Los ángulos i_h y la diferencia $\phi - \phi_s$ corresponden al ángulo de salida del rayo y el acimut de éste con respecto al de la traza de la falla; ϕ_s , δ y λ corresponden al acimut, echado y dirección de movimiento relativo de la falla. Para un medio homogéneo, el rayo sale de la fuente como el vector γ . Los vectores unitarios $\hat{i}$ (componente longitudinal, onda P); $\hat{\phi}$ (componente transversal tangencial, onda SH) y $\hat{p}$ (componente transversal vertical, onda SV) describen componentes del desplazamiento sísmico en la superficie. La figura corresponde a la 4.20 de Aki y Richards (1980).

MEDIO HOMOGÉNEO INFINITO

Las gráficas de ejemplo del Programa 1 aparecen en las figuras 2 (onda P) y 3 (onda S). Las variables están normalizadas según la ecuación (4.33) de Aki y Richards (1980); el cálculo no necesita parámetros de entrada. Para el patrón de radiación de la onda S, hemos tomado la opción de “cortar” la superficie cerrada variando el ángulo acimutal sólo en 180°. El lector puede, como ejercicio, determinar la dirección del doble par de fuerzas que, en el origen de coordenadas, equivalen a la dislocación puntual. Estas figuras son las que más aparecen en la literatura (v. gr., Kennet, 1983, p. 90); el usuario puede modificar el programa y obtener dibujos con el patrón de radiación de otros modelos de fuentes puntuales simples (explosión, monopolo, etc.).

Programa 1. Programa para calcular y visualizar el patrón de radiación de una fuente puntual (dislocación de cizalle), ondas P y S, en un medio homogéneo e infinito; campo lejano. No se incluyen factores $1/\alpha^3$ y $1/\beta^3$ en cada patrón para el caso de onda P y S, respectivamente; α y β son las velocidades sísmicas correspondientes. Los factores producen una amplificación relativa de 5.2 para las amplitudes de onda S con respecto a las de P si el material es poissoniano. Los factores no se incluyen en éste y otros casos para que las figuras muestren una mejor resolución en los detalles. Otros factores tampoco incluidos son iguales para ambos patrones.

```
(* Patrones de Radiación de Ondas P y S
en un Medio Homogéneo Infinito *)

(* P: las tres componentes en coordenadas cartesianas a partir
de las esféricas {t: ang. polar; p: ang. acimutal} *)

radP[t_,p_]:= (Sin[2 t] Cos[p]){
  Sin[t] Cos[p], Sin[t] Sin[p], Cos[t]}
(* Comando que evalúa y grafica la expresión anterior *)
ParametricPlot3D[
  radP[t,p]//Evaluate,
  {t, 0, Pi}, {p, 0, 2Pi},
  (* Boxed->False, Axes->None, *)
  (* AspectRatio -> Automatic, *)
  PlotPoints->{25,25},
  ViewPoint -> {1,1,0.5}];

(* S *)
radS[t_,p_]:= (Cos[2 t]^2 Cos[p]^2 +
  Cos[t]^2 Sin[p]^2)^(1/2){
  Sin[t] Cos[p], Sin[t] Sin[p], Cos[t]}
ParametricPlot3D[
  radS[t,p]//Evaluate,
  {t, 0, Pi}, {p, Pi, 2Pi},
  (* Boxed->False, Axes->None, *)
  (* AspectRatio -> Automatic, *)
  PlotPoints->{25,25},
  ViewPoint -> {1,1,0.5}];
```

Las figuras muestran claramente ejes de simetría de las superficies graficadas; de ahí se deduce la no-unicidad para elegir el plano de falla entre los dos planos de simetría con valores nulos de amplitud de la onda P. Estos planos se designan como planos nodales en Sismología.

SEMIESPACIO LATERALMENTE HOMOGÉNEO

El Programa 2 calcula las amplitudes de los arribos P, SV y SH en sus componentes horizontales y verticales. Los valores numéricos de los parámetros más generales (λ : dirección del desplazamiento relativo en la falla con respecto a la dirección horizontal; δ : echado con respecto al plano horizontal) corresponden a un fallamiento de rumbo ($\lambda = 0^\circ$, $\delta = 90^\circ$). En este programa, he fijado las velocidades de V_p en la fuente y en la superficie libre como 5.2 km/s y 6.2 km/s, respectivamente, y la razón de velocidad $V_p/V_s = \sqrt{3}$. Valores semejantes son típicos en estructuras continentales como en la Sierra Peninsular del norte de Baja California (Nava y Brune, 1982). Las convenciones para ángulos que describen fallamiento, ejes de co-




Figura 2. Patrón de radiación de onda P que resulta del programa mostrado en el Programa 1.




Figura 3. Patrón de radiación de onda S que resulta del programa mostrado en el Programa 1.

Programa 2. Programa para calcular y visualizar el patrón de radiación (campo lejano) de una fuente puntual (dislocación de cizalle) en puntos de la superficie libre de un medio semi-infinito lateralmente homogéneo. Ver detalles en el texto. Se calculan los arribos P-vert, P-hor, SV-hor, SV-vert y SH, donde la nomenclatura de cada arribo indica la onda incidente y la componente del desplazamiento. Cada arribo contiene el campo incidente y el reflejado, incluyendo ondas convertidas. Ver comentarios sobre normalización en el pie de la Figura 2. Notación para los parámetros de entrada: velocidades en el foco (*alphah*, *betah*) y en la superficie (*alpha0* y *beta0*), ángulo de desplazamiento en la falla (*lamb*) y echado de la falla (*del*); además, como argumentos en el comando ParametricPlot3D: *in*, ángulo de salida del rayo desde la fuente; *dphi*: acimut del rayo con respecto a la dirección de la falla.

```
(* Prad_tod.Ma: Patrones de Radiación en Superficie Libre
P-vert, P-hor, SV-vert, SV-hor y SH.
Faltan factores (1/Vp)^3, (1/Vs)^3 en las amplitudes.
Ref. Aki&Richards, eqs. (4.84), ((4.85), (4.86), (4.87) y
probl. 5.6 *)

(* Parámetros de entrada *)
vpvs=Sqrt[3];
gtorad = Pi/180;
lamb = 90 gtorad;
del = 45 gtorad;
alphah = 6.2;
betah = alphah/vpvs;
alpha0 = 5.2;
beta0 = alpha0/vpvs;

palph = Sin[ih] / alphah;
pbeta = Sin[ih] / betah;

(* Coordenadas cartesianas *)
xP = Sin[ih] Cos[dphi];
yP = Sin[ih] Sin[dphi];

(* Proyeccion equiareal
xP = Sqrt[2] Sin[ih/2] Cos[dphi];
yP = Sqrt[2] Sin[ih/2] Sin[dphi]; *)

(* P *)
sinPP0 = palph alpha0;
iPP0 = ArcSin[sinPP0];
sinPSV0 = palph beta0;
iPSV0 = ArcSin[sinPSV0];
denomP = (1.0/beta0^2 - 2 palph^2)^2 +
4 palph^2 Cos[iPP0] Cos[iPSV0]/(alpha0 beta0);

t1p = Cos[lamb] Sin[del] Sin[ih]^2;
t2p = -Cos[lamb] Cos[del] Sin[2 ih];
t3p = Sin[lamb] Sin[2 del];
t4p = Sin[lamb] Cos[2 del] Sin[2 ih];
rP = Sin[2 dphi] t1p +
Cos[dphi] t2p +
(Cos[ih]^2 - Sin[ih]^2 Sin[dphi]^2) t3p +
Sin[dphi] t4p;

(* P-vert *)
dPvert = -2 rP Cos[iPP0] *
(1/beta0^2 - 2 palph^2) /
(beta0^2 denomP);
dPv = (dPvert^2)^0.5;
ParametricPlot3D[{xP, yP, dPv},
{ih,0,Pi/2},{dphi,-Pi,Pi},
PlotPoints->{60,40},
PlotLabel->"(P)vert: delta = 450, lambda = 900",
AxesLabel->{"Fault Trace"," "," "};]

(* P-hor *)
dPhor = 4 rhoP palph Cos[iPP0] Cos[iPSV0] /
(beta0^3 denomP);
dPh = (dPhor^2)^0.5;
ParametricPlot3D[{xP, yP, dPh},
{ih,0,Pi/2},{dphi,-Pi,Pi},
PlotPoints->{60,40},
PlotLabel->"(P)hor: delta = 450, lambda = 900";]

(* SV *)
senSVSV0 = pbeta beta0;
iSVSV0 = ArcSin[senSVSV0];
ciSVP0 = Abs[1.0-pbeta^2 alpha0^2]^0.5;
t12 = (1.0/beta0^2 - 2 pbeta^2)^2;
t22 = 4 pbeta^2 Cos[iSVSV0] ciSVP0/(alpha0 beta0);
denomSV = If[alpha0^2 pbeta^2 <= 1,
t12 + t22,
Sqrt[t12^2 + t22^2]];

t1sv = Sin[lamb] Cos[2 del] Cos[2 ih];
t2sv = -Cos[lamb] Cos[del] Cos[2 ih];
t3sv = 0.5 Cos[lamb] Sin[del] Sin[2 ih];
t4sv = -0.5 Sin[lamb] Sin[2 del] Sin[2 ih];
rSV = Sin[dphi] t1sv +
Cos[dphi] t2sv +
Sin[2 dphi] t3sv +
(1+Sin[dphi]^2) t4sv;
rhoSV = (rSV^2)^0.5;

(* SV-hor *)
dSVhor = 2 rhoSV Cos[iSVSV0] *
(1/beta0^2 - 2 pbeta^2) /
(beta0^2 denomSV);
dSVh = (dSVhor^2)^0.5;
ParametricPlot3D[{xP, yP, dSVh},
{ih,0,Pi/2},{dphi,-Pi,Pi},
PlotPoints->{60,40},
PlotLabel->"(SV)hor: delta = 450, lambda = 900";]
(* AxesLabel->{"Fault Trace"," "," "}; *)

(* SV-vert *)
dSVver = 4 rhoSV pbeta Cos[iSVSV0] ciSVP0 /
(alpha0 beta0^2 denomSV);
dSVv = (dSVver^2)^0.5;
ParametricPlot3D[{xP, yP, dSVv},
{ih,0,Pi/2},{dphi,-Pi,Pi},
PlotPoints->{60,40},
PlotLabel->"(SV)ver: delta = 450, lambda = 900";]
(* AxesLabel->{"Fault Trace"," "," "}; *)

(* SH *)
t1sh = Cos[lamb] Cos[del] Cos[ih];
t2sh = Cos[lamb] Sin[del] Sin[ih];
t3sh = Sin[lamb] Cos[2 del] Cos[ih];
t4sh = -0.5 Sin[lamb] Sin[2 del] Sin[ih];
rSH = Sin[dphi] t1sh +
Cos[2 dphi] t2sh +
Cos[dphi] t3sh +
Sin[2 dphi] t4sh;
rhoSH = 2 (rSH^2)^0.5;
ParametricPlot3D[{xP, yP, rhoSH},
{ih,0,Pi/2},{dphi,-Pi,Pi},
PlotPoints->{60,40},
PlotLabel->"SH: delta = 450, lambda = 900";]
(* AxesLabel->{"Fault Trace"," "," "}; *)
```

ordenadas relacionadas con patrones de radiación y signos en los coeficientes de reflexión y transmisión son de Aki y Richards (1981). (Ver figuras 1 de este artículo y 5.5, p. 139, de ese texto). Para distinguir el cambio de signo, se grafica el valor absoluto de las amplitudes. El cambio de signo del primer arribo P, es decir, de compresivo (positivo: de la fuente hacia la estación) a dilatacional (viceversa) está, por la tanto, marcado por la discontinuidad que toca el plano ecuatorial en la Figura 4; ellas delimitan la distribución de compresiones y dilataciones en el caso de falla normal (*dip-slip fault*, en inglés).

Las figuras 4 (P-vert), 5 (P-hor), 6 (SV-hor), 7 (SV-vert) y 8 (SH) muestran patrones de radiación en la superficie libre para una falla normal ($\lambda = 90^\circ$; $\delta = 45^\circ$). En los paréntesis anteriores hemos identificado los arribos por el tipo de onda incidente. El caso “SV-vert” es importante porque contiene el caso de reflexión supercrítica para la onda reflejada (convertida) P; ello corresponde a un ángulo de reflexión imaginario y a valores complejos del coeficiente de reflexión. Aquí, el valor absoluto de la amplitud corresponde al módulo del número complejo correspondiente. La Figura 6 muestra la discontinuidad resultante. No me pareció interesante incluir gráficas con la distribución del cambio de fase asociado.

El Programa 3 calcula la razón de amplitudes (P/SV)-vert; esta razón se ha propuesto como herramienta para determinar mecanismos focales con datos locales (Kisslinger, 1980; Kisslinger *et al.*, 1981; ver también una corrección en Kisslinger, 1982). En vez de aplicar la ecuación (corregida) de Kisslinger (1982), simplemente se calcula la razón entre las amplitudes ya referidas en el programa. La Figura 9 entrega el resultado para el caso de un fallamiento de rumbo donde ocurre una división de dos números que tienden a cero (en los puntos correspondientes a los planos nodales), por lo que debemos tomar precauciones. Para una simple división por cero o un número muy pequeño, podríamos reemplazar a este valor por uno suficientemente pequeño (0.0001, por ejemplo); sin embargo, este artefacto no produce necesariamente el límite de la razón entre dos valores que tienden a cero. Es más exacto buscar matemáticamente el límite e introducirlo como opción en el programa cuando el caso especial lo amerite. Sin embargo, dado que la discontinuidad que se puede formar se encuentra en direcciones muy específicas (la de los ejes de coordenadas, en el caso que nos preocupa), se puede utilizar la discretización y la continuidad probada analíticamente para salir más fácilmente del paso. En pocas palabras, se evita hacer el cálculo sobre los puntos que presentan la singularidad removible; para ello, se mueve ligeramente el rango de cálculo del acimut con respecto a la dirección de la falla.

La Figura 9 muestra que la razón (SV/P)-vert no tiene resolución para determinar los planos nodales en un fallamiento de rumbo, resultado ya comentado en Kisslinger (1980). Invitamos al usuario a tratar el caso en que $\lambda = 90^\circ$ y $\delta = 45^\circ$, para el cual las amplitudes P-vert y SV-vert están representadas en las Figuras 5 y 7. Las simetrías de la superficie resultante hacen poco posible la determinación de los planos nodales con este




Figura 4. Patrón de radiación del arribo P-vert en la superficie libre, según el programa mostrado en el Programa 2. Se trata de una fallamiento normal ($\lambda = 90^\circ$; $\delta = 45^\circ$).




Figura 5. Igual que en la Figura 4, para el arribo P-hor.




Figura 6. Igual que en la Figura 4, para el arribo SV-hor.




Figura 7. Igual que en la Figura 4, para el arribo SV-vert.




Figura 8. Igual que en la Figura 4, para el arribo SH.

tipo de información; además, la superficie puede ofrecer una gran complejidad en conducta, con hojas que suben y bajan con gran amplitud y en un rango pequeño de las variables independientes.

En resumen, la Figura 9 y el ejercicio que recomendamos indican que la información de la razón (SV/P)-vert no es adecuada para los propósitos de determinar mecanismos focales cuando los fallamientos de rumbo y normal son predominantes en el área de estudio. Esto es debido, además, a que la indeterminación de la estructura hace que sea insegura la posición en el plano de las fuertes discontinuidades producidas por la aparición de ondas inhomogéneas. El usuario puede modelar la presencia de sedimentos en la superficie, con $V_p = 2.5$ km/s y observar el cambio en los resultados. En los programas descritos en esta sección, utilizo $i_h = [0 - \pi/2]$; el usuario puede tomar $i_h > \pi/2$ en proyección equiareal teniendo los cuidados que se requieran. Los ejemplos mostrados en esta sección fueron diseñados teniendo en mente la interpretación de datos locales.

Por último, el Programa 4 sirve para hacer dibujos de funciones univariantes que corresponden a perfiles de las figuras tridimensionales para un acimut constante. Con ellas, por ejemplo, se puede duplicar las amplitudes mostradas en la Figura 5.10 del texto de Aki y Richards (1981).

CONCLUSIONES

Hemos entregado programas en el lenguaje de *Mathematica* para cálculos del patrón de radiación tanto en un medio infinito como en la superficie libre de un medio semiinfinito lateralmente homogéneo. No se incluyen efectos de atenuación que afectan mayormente a las amplitudes de onda incidentes S (SV o SH) con respecto a las de P y S. Aún así, consideramos que las gráficas son útiles en la docencia, en trabajos de interpreta-

Programa 3. Programa para calcular y visualizar los arribos P-vert, SV-vert y la razón de amplitudes (SV/P)-vert. Ver comentarios en el pie de la Figura 5 para explicaciones generales. Note que hemos movido ligeramente el rango de una variable independiente para calcular –en el comando ParametricPlot3D– la razón de amplitudes; ésto, con el objeto de resolver una singularidad removible en el caso de un fallamiento de rumbo.

```
(* Radsv_p.Ma: Patron de Radiacion para P-vert, SV-vert
y (SV/P)-vert *)

(* Parametros generales y de entrada *)
gtorad = Pi/180;
lamb = 0 gtorad;
del = 90 gtorad;
vpvs = Sqrt[3];
alphah = 6.2;
betah = alphah/vpvs;
alpha0 = 5.2;
beta0 = alpha0/vpvs;

(* Se usa para evitar singularidad en discretizacion *)
eps = 0.0012345;

palph = Sin[ih] / alphah;
pbeta = Sin[ih] / betah;

(* Ejes de coordenadas cartesianas *)
xP = Sin[ih] Cos[dphi];
yP = Sin[ih] Sin[dphi];

(* P *)
sinPP0 = palph alpha0;
iPP0 = ArcSin[sinPP0];
sinPSV0 = palph beta0;
iPSV0 = ArcSin[sinPSV0];
denomP = (1.0/beta0^2 - 2 palph^2)^2 +
4 palph^2 Cos[iPP0] Cos[iPSV0]/(alpha0 beta0);

t1p = Cos[lamb] Sin[del] Sin[ih]^2;
t2p = -Cos[lamb] Cos[del] Sin[2 ih];
t3p = Sin[lamb] Sin[2 del];
t4p = Sin[lamb] Cos[2 del] Sin[2 ih];
rP = Sin[2 dphi] t1p +
Cos[dphi] t2p +
(Cos[ih]^2 - Sin[ih]^2 Sin[dphi]^2) t3p +
Sin[dphi] t4p;
rhoP = (rP^2)^0.5;

(* P-vert *)
dPvert = 2 rhoP alpha0 Cos[iPP0] *
(1/beta0^2 - 2 palph^2) /
(beta0^2 alpha0 denomP);
dPv = (dPvert^2)^0.5;

(* SV *)
senSVSV0 = pbeta beta0;
iSVSV0 = ArcSin[senSVSV0];
ciSVP0 = Sqrt[Abs[1.0-pbeta^2 alpha0^2]];
t12 = (1/beta0^2 - 2 pbeta^2)^2;
t22 = 4 pbeta^2 Cos[iSVSV0] ciSVP0/(alpha0 beta0);
denomSV = If[alpha0^2 pbeta^2 <= 1,
t12 + t22,
Sqrt[t12^2 + t22^2]];

t1sv = Sin[lamb] Cos[2 del] Cos[2 ih];
t2sv = -Cos[lamb] Cos[del] Cos[2 ih];
t3sv = 0.5 Cos[lamb] Sin[del] Sin[2 ih];
t4sv = -0.5 Sin[lamb] Sin[2 del] Sin[2 ih];
rSV = Sin[dphi] t1sv +
Cos[dphi] t2sv +
Sin[2 dphi] t3sv +
(1+Sin[dphi]^2) t4sv;
rhoSV = (rSV^2)^0.5;

(* SV-vert *)
dSVver = 2 rhoSV pbeta Cos[iSVSV0] ciSVP0 /
(alpha0 beta0^2 denomSV);
dSVv = (dSVver^2)^0.5;

(* (SV/P)-vert *)
ratio = dSVv/dPv;

(* Llamados para calculo y graficado *)
ParametricPlot3D[{xP, yP, dPv},
{ih,0,Pi/2},{dphi,-Pi,Pi},
PlotPoints->{60,40},
PlotLabel->"(P)vert: delta = 900, lambda = 00",
AxesLabel->{"Fault Trace"," "," "};

ParametricPlot3D[{xP, yP, dSVv},
{ih,0,Pi/2},{dphi,-Pi,Pi},
PlotPoints->{60,40},
PlotLabel->"(SV)ver: delta = 900, lambda = 00";
(* AxesLabel->{"Fault Trace"," "," "}; *)

ParametricPlot3D[{xP, yP, ratio},
(* Para falla normal: {ih,0,0.5 Pi/2} *)
{ih,eps,Pi/2 - eps},{dphi,-Pi + eps,Pi + eps},
PlotPoints->{60,40},
PlotLabel->"(SV/P)vert: delta = 900, lambda = 00";
(* AxesLabel->{"Fault Trace"," "," "}; *)
```

ción rutinaria y aún en investigación, además de que es una buena forma de introducirse en el uso de *Mathematica*.

AGRADECIMIENTOS

Agradezco a Victor M. Frías C. por su ayuda técnica para poner al día mi PC, a Ruth Eaton por su ayuda en la edición y a los revisores por mejoramientos en la exposición.

Programa 4. Programa para calcular y visualizar un perfil, para un acimut constante, de la amplitud SV, componentes vertical y horizontal, en función del seno del ángulo de salida desde la fuente puntual. Este programa es útil para definir el rango en que el apareamiento de reflexiones P supercríticas produce una fuerte variación del patrón de radiación resultante.

(* RadsV1d.Ma: Perfil, para acimut constante, del patrón de radiación en la superficie libre de un medio semiinfinito y lateralmente homogéneo *)

(* Parámetros generales y de entrada *)
 gtorad = Pi/180;
 alphah = 6.20;
 betah = alphah/1.73;
 alpha0 = 5.20;
 beta0 = alpha0/1.73;
 lamb = 90 gtorad;
 del = 45 gtorad;
 dphi = 45 gtorad;
 eps = 0.00012345;

palph = Sin[ih] / alphah;
 pbeta = Sin[ih] / betah;
 xP = Sin[ih];

(* SV *)
 senSVSV0 = pbeta beta0;
 iSVSV0 = ArcSin[senSVSV0];
 ciSVP0 = Sqrt[Abs[1.0-pbeta^2 alpha0^2]];
 t12 = (1.0/beta0^2-2 pbeta^2)^2;
 t22 = 4 pbeta^2 Cos[iSVSV0] ciSVP0/(alpha0 beta0);
 denomSV = If[alpha0^2 pbeta^2 <= 1,
 t12 + t22,
 Sqrt[t12^2 + t22^2]];

t1sv = Sin[lamb] Cos[2 del] Cos[2 ih];
 t2sv = -Cos[lamb] Cos[del] Cos[2 ih];
 t3sv = 0.5 Cos[lamb] Sin[del] Sin[2 ih];
 t4sv = -0.5 Sin[lamb] Sin[2 del] Sin[2 ih];
 rSV = Sin[dphi] t1sv +
 Cos[dphi] t2sv +
 Sin[2 dphi] t3sv +
 (1+Sin[dphi]^2) t4sv;
 rhoSV = (rSV^2)^0.5;

(* SV-hor *)
 dSVhor = 2 rhoSV Cos[iSVSV0]*
 (1/beta0^2 - 2 pbeta^2) /
 (beta0^2 denomSV);
 dSVh = (dSVhor^2)^0.5;

ParametricPlot[{xP,dSVh},
 {ih,0,Pi/2},
 PlotPoints -> 200,
 PlotLabel -> "(SV)hor: delta=450, lambda=900, dphi=450"];

(* SV-vert *)
 dSVver = 4 rhoSV pbeta Cos[iSVSV0] ciSVP0 /
 (beta0^2 alpha0 denomSV);
 dSVv = (dSVver^2)^0.5;
 ParametricPlot[{xP,dSVv},
 {ih,0-eps,Pi/2 - eps},
 PlotPoints -> 200,
 PlotLabel -> "(SV)ver: delta=450, lambda=900, dphi=450"];




Figura 9. Razón de la componente vertical de amplitudes (SV/P) para fallamiento puro de rumbo. La superficie graficada muestra claramente que esta razón no contiene información para determinar la posición de los planos nodales en este caso.

REFERENCIAS

- Aki, K. y P. Richards (1981) Quantitative Seismology, W.H. Freeman, San Francisco, California, E.U.A., 932 pp.
- Cordero, L.A., M. Fernández y A. Gary (1995) Geometría Diferencial de Curvas y Superficies, Addison-Wesley Iberoamericana, Wilmington, Delaware, E.U.A., 823 pp.
- Crandall, R.E. (1991) Mathematica for the Sciences, Addison-Wesley Pub. Co., Inc. Redwood City, Ca, E.U.A., 300 pp.
- Frez, J. (1997) Geofísica computacional y enseñanza en el Posgrado. GEOS, Unión Geofísica Mexicana, vol. 17, no 2, p. 60-66.
- Herrman, R.B. (1975) A student's guide to the use of P and S wave data for focal mechanism determination, Earthquake Notes, vol. 46, pp. 29-39.
- Jarosh, H. y E. Aboodi (1970) Towards a unified notation of source parameters, Geophysical Journal Royal Astronomical Society, vol. 21, pp. 513-529.
- Jost, M. L. y R.B. Herrman (1989) A student's guide to a review of moment tensors, Seismological Research Letters, vol. 60, pp. 37-57.
- Kisslinger, C. (1980) Evaluation of S to P amplitude ratios for determining focal mechanisms from regional network observations, Bulletin of the Seismological Society of America, vol. 70, pp. 999-1014.
- Kisslinger, C., J.R. Bowman y K. Koch (1981) Procedures for computing focal mechanisms from local (SV/P) data, Bulletin of the Seismological Society of America, vol. 71, 1719-1729.

- Kisslinger, C. (1982) Errata, Evaluation of S to P amplitude ratios for determining focal mechanisms from regional network observations, Bulletin of the Seismological Society of America, vol. 72, p. 344.
- Pujol, J., y R.B. Herrman (1990) A student guide to point sources in homogeneous media, Seismological Research Letters, vol. 61, pp. 209-224.
- Nava, F. A., y J. Brune (1982) An earthquake-explosion reversed refraction line in the Peninsular Ranges of southern California and Baja California norte, Bulletin of the Seismological Society of America Bulletin, v. 72, p. 1195-1206.
- Simila, G. W. (1983) Computer application in undergraduate geophysics: fault-plane solutions, body-wave radiation patterns and amplitude ratios, Computers and Geosciences, vol. 8, pp 65-76.
- Wolfram, S. (1991) *Mathematica*, A System for Doing Mathematics by Computer, Addison-Wesley Publishing Co., 961 pp.